page 1 NITIATION VERILOG page 1

10.

11

INITIATION AU LANGAGE VERILOG

Jean Louis Noullet - 1995-2005 - 5eéme édition rev.5

. Introduction- - - - - - - - - - - - - - - - - oo oo 2
1.1 Objectifs 2
1.2 Les supports de données en Verilog 3
. Des exemples, pour commencer - - - - - - - - - - - - 3
2.1 Simulation d’un “latch RS” 3
2.2 Un exemple de conception descendante ("TOP-DOWN") 6
2.3 Simulation d’un circuit synchrone 8
. Déclarations - - - - - - - - - - - - - - - - - - - - 10
. Expressions et Constantes - - - - - - - - - - - - - - 11
4.1 opérations logiques “verticales” ou “bitwise” 11
4.2 opérations logiques “horizontales” ou “unary” 11
4.3 opérations arithmétiques et décalage 11
4.4 opération booléennes 12
4.5 opérations relationnelles 12
4.6 vecteurs 12
4.7 constantes numériques 12
4.8 extension - troncature 13
. Communication reg <-->net- - - - - - - - - - - - - 13
.EtatsetForces - - - - - - - - - - - - - - - - - - - 14
. Blocs Procéduraux - - - - - - - - - - - - - - - - - 15
7.1 Regles 15
7.2 Temporisation procédurale 16
. Constructions Procédurales - - - - - - - - - - - - - 17
8.1 Itérations ou boucles (for, while, repeat, forever) 17
8.2 Décisions (if, expr. conditionnelle, case) 18
. Encore quelques exemples- - - - - - - - - - - - - - 19
9.1 bascule D avec clear asynchrone 19
9.2 Bulffer tri-state 20
9.3 Cellule paramétrée 20
Introduction a la Synthese Logique - - - - - - - - - 21
10.1 Modules combinatoires 21
10.2 Modules séquentiels synchrones 23
10.3 Machine a états finis 24
10.4 Registre avec clear asynchrone 25
.Cequiresteraitavoir - - - - - - - - - - - - - - - 26

© J.L.N.2005



page 2 NITIATION VERILOG page 2

| NITIATION AU LANGAGE VERILOG

1. Un langage de description de matériel

1.1 Objectifs

Les langages de description de matérldD{. = HardwareDescriptionLangage) font partie des

outils de base pour la conception de systemes logiques intégrés cablés, que le produit final soit
construit sur des composants électriguement configurablR&A = Field Programmablel ogic

Array) ou des circuits intégrés spécifiqudsS(C = Application Specific | ntegratedCircuit).

Ces langages doivent étre utilisables pour :

» écrire laspécificationd’'un systeme (descriptiocomportementaleu fonctionnellg

 le construire par interconnexion de cellules élémentaires (descriginrcturelleou physiqué
» gérer des descriptioniérarchiséeqdont la description structurelle est un cas patrticulier)
 valider parsimulation les différents types de description

» permettre l'utilisation de programmes dgnthése automatique

De plus les descriptions comportementales doivent se faire dans deux modes :

* modeconcurrent: les données se propagent dans des élements dont le comportement est décrit
par des déclarations qui ont un effet permanent (comme dans les descriptions structurelles)

* modeprocédural: les données sont manipulées par des séquences d’instructions (comme dans
les programmes d’ordinateurs)

Dans le mode concurrent les données sont actives, on pattatieflow”, alors que dans le mode
procédural c’est le programme qui contrdle les évenements, on parle algmodeam flow”.

Il existe deux standards de langages de description de matériel :

 VHDL

» Verilog

Leurs syntaxes sont assez différentes mais mettent en oeuvre les mémes concepts.

1.2 Les supports de données en Verilog

Le langage Verilog utilise deux classes de supports de données distinctes pour les modes
concurrents et procedural.

» classe&net” dans le mode concurrent : un net est équivalent a un noeud d’un circuit électrique,
son état est contrélé en permanence par les éléments aux sorties desquels il est connecté.

» class€reg” dans le mode procédural : un reg est équivalent a une variable d’'un programme
informatique : il subit des affectations instantanées par instructions et conserve son état jusqu’a

la prochaine affectation.

Le bon usage des nets et regs est la seule réelle difficulté de Verihoga@onsacré le chapitée

2. Des exemples, pour commencer

2.1 Simulation d’'un “latch RS”

bibliotheque

Y

stimuli — circuit — | interpretation

environnement de simulation

Dans cette exemple, on utilisera le langage Verilog pour décrire un petit circuit, la bibliotheque de
cellules standard qui sert a le construire, et I'environnement de simulation qui sert & le valider.

2.1.1 Description structurelle de la bascule rs_latch :

module rs_latch( R, S, Y ); R
output Y; _Y
wire YB;

input R, S; ><

LibNo2 11 (Y, R, YB); (YB)
LibNo2 12 (YB, Y, S); S

endmodule

© J.L.N.2005

Un moduleou cellule est un noeud de la hiérarchie structurelle.

Sa description commence par la déclarationptess qui servent & le connecter au monde extérieur
(les ports sont de la classeet").

Suivent desinstancesou "appels" de modules situés en dessous dans la hiérarchie. A chaque
instance est associé un nom d’instance unique (ici I1 et 12).

Ici, la cellule LibNo2 est supposée faire partie d'une bibliotheque de cellules standard.

Dans ce module, un net interne YB est déclaré en tantquee(classe'net").

Les identifieurs créés dans le module (R, S, Y, YB, 11, 12) sont locaux.

La connexion des nets de rs_latch aux ports de chaque instance de LibNo2 est régreigadds
éléments dans la liste. Le langage permet une variante : la connexion par noms, plus explicite mais
plus lourde :

module rs_latch(R, S, Y);
output Y; input R, S;
wire YB;
LibNo2 I1 ( .a(R), .b(YB), .s(Y));
LibNo2 12 ( .s(YB), .a(Y), .b(S));
endmodule

2.1.2 Description comportementale des cellules standard

Pour simuler notre rs_latch, nous avons besoin d'une description comportementale de LibNo2 :

“timescale 1ns/10ps
module LibNo2 (s, a, b); // porte NOR
output s; input a, b;

assign #(2.5, 2.0) s = ~(a|b);
endmodule

La directive de compilationtimescalesert a indiquer qu'on veut donner les temps en ns avec une
résolution de 10 ps.

L'instructionassignétablit une connexion permanente ("data-flow") entre I'expresdiajb) et

le nets, avec un délai de 2.5 ns a la montée et 2 ns a la descente.

(La syntaxe des expressions est détaillée dans le chdpitre

2.1.3 Environnement de simulation

L'environnement de simulation de notre circuit devra assurer la génération des stimuli et I'affichage
des résultats, ce qu'on fera de maniere procédurale.

Ces éléments doivent étre hébergés dans un module dit module de test qui constituera la racine de la
hiérarchie. Ce module n’a ni entrée ni sortie.

Pour que les signaux injectés dans le circuit (stimuli) puissent étre contrblés de maniere

procédurale, le module de test doit contenir des regs connectés aux entrées du circuit.
© J.L.N.2005

Par contre les sorties du circuit doivent étre connectées a des nets (déclarés comme “wire”).

module test;
regr, s; wirey;
rs_latch lat(r, s, y); // instance de notre circuit
initial
begin
$display("simulation RS");
$monitor("t = %4g ns : rs = %b%b --> %b (%b)",
$realtime, 1, s, y, lat.YB);
r=0;s=0;
#51 =1,
#5r1=0;
#5s=1,;
#5s=0;
#5s=1;r=1,
#5 $display("t=%g ns : travail fini", $realtime );
end
endmodule

Apres ces déclarations de signaux, le module test contient une instance du circuit rs_latch lui-
méme, et enfin ubloc procédural(initial begin ......... end)

Dans ce bloc procédural, il y a une séquence dinstructions d'affectation dont certaines sont
précedées d'unsondition de temporisatio#5 exprimant une attente de 5 ns.

Les identifieurs commencant par $ désignent des fonctions ou taches prédéfinies dans le simulateur
Verilog-XL.

$display  affiche du texte a l'instant ou il est exécuté, alors §meonitor déclenche un
processus automatique qui va réitérer l'affichage chaque fois qu'un des signaux concernés change
d'état.

L'expressionlat.YB  permet de désigner un signal situé "en dessous" dans la hiérarchie en
indiquant un chemin constitué de noms d’instances.

Dans cet exemple tres simple, on a rencontré les 3 niveaux de description :

- structurelle (rs_latch)

- comportementale "data-flow" (LibNo2)

- procédurale (test)

Le langage permet de mélanger librement les 3 niveaux.

N.B. :

» Toute description doit étre hébergée par un module

» Toute description procédurale doit étre hébergée par un bloc procéuhitial” ou "always"
(chapitre7.).

2.2 Un exemple de conception descendante ("“TOP-DOWN")

Un registre a décalage a chargement parallele de 4 bits ("PISQO", pour "Parallel-In, Serial-Out").

ser_out
PISO —

A
Ioadl} ?clk par_in

Le signal load commande le chargement d’'un mot par_in (ici de 4 bits) que le registre devra
restituer bit par bit sur la sortie ser_out, en commencant par le bit de faible poids, avec un
fonctionnement strictement synchrone (horloge clk).

Considérons une conception manuelle, partant d'une spécification consistant en une description
comportementale globale (82.2.1), et aboutissant a une structure basée sur des cellules standard
disponibles (82.2.3), avec une étape intermédiaire de découpage en “étages” (§82.2.2).

2.2.1 Modele global du registre en Verilog :

/I description comportementale du registre piso
module piso4( clk, load, par_in, ser_out);
input clk, load; input [3:0] par_in;
output ser_out;
reg [3:0] contenu;
always @ (posedge clk)
begin
if (load)
contenu = par_in; /I chargement
else
contenu = contenu >> 1; // decalage
end
assign #3 ser_out = contenul0];
endmodule

L'entrée parallelgoar_in  de ce registre est un groupe de 4 connexions représentées par un seul
nom. Cette maniere de désigner les signaux comme des tableaux indicés est souventiapgelée

ou mieux :"vecteur".

La déclaration comprend les valeurs extrémes de lindice entre crochets. Le premier indice
correspondant au bit de fort poids, il est vivement conseillé de toujours mettre le plus fort indice en
premiere position.

Quelques exemples d'utilisation :

contenu[0] estle"Isb” (less significant bit = bit de plus faible poids) dentenu ,

contenu[2]  est son bit de poids 4, © J.L.N.2005

par_in[2:0] est un bus de 3 bits contenant lelsBdepar_in
par_in[3:0] est synonyme dpar_in

Le coeur du modéle est un bloc procédwatvays dont la condition de déclenchement est le front
montant de I'horlogelk . (cf chapitre7.)

La mémoire interne du registre est modélisée par lecasgenu . La sortie sérieser_out est
connectée de fagon permanente au bit de faible poid®nnu par une directive assign.

2.2.2 Modeéle découpé en étages :

TOISNUTT IS IS Iy

[t

par_in[3] par_in[2] par_in[1] par_in[0O]

/I description structurelle du registre piso

module piso4( clk, Id, par_in, ser_out);

input clk, load; input [3:0] par_in;

output ser_out; wire q1, g2, g3;
etage e3 ( clk, load, par_in[3], 0, g3 );
etage e2 ( clk, load, par_in[2], 93, 92 );
etage el ( clk, load, par_in[1], g2, q1);
etage e0 ( clk, load, par_in[0], g1, ser_out);

endmodule

N.B. g3, g2, gql1 sontdes nets internes de piso4. On a relig(aéro) I'entrée de décalage de
I'étage e3 pour obtenir un comportement semblable a celui de I'opératerr de la description
comportementale.

/I decription comportementale d'un etage de registre
module etage ( clk, select, d1, dO, q);
input clk, select, d1, dO; output q;

regrq; "
always @ (posedge clk) —o\ q
if (select) rq =d1; d1_ A
else rg = do; |
assign #1 g = rq; select  clk
endmodule

© J.L.N.2005

2.2.3 Modele niveau cellules standard

/I decription structurelle d'un etage de registre
module etage ( ck, select, d1, dO, q);
input ck, select, d1, dO; output q; wire d;

LibDFF bascule ( .q(q), .ck(ck), .d(d) );

LibMux multipl ( .y(d), .i0(d0), .i1(d1), .s(select) );
endmodule

Ceci suppose l'existence des cellules standard LibDFF et LibMux, dont les descriptions
comportementales pourraient étre :

/l bascule D sensible au front montant
module LibDFF( g, ck, d);
input ck, d; output q;
reg resu;
always @ ( posedge ck ) resu =d;
assign #1 q = resu;
endmodule

/I multiplexeur & deux entrées

module LibMux(i0, i1, s,y );

input i0, i1, s; output y;
assign#1ly=(i10&~s) || (i1 &s);

endmodule

2.3 Simulation d’un circuit synchrone

On va profiter du parallélisme offert par le langage pour produire le signal d’horloge périodique
avec une boucle séparée de la séquence de test principale.

D’autre part on doit impérativement éviter toute collision entre les transitions des entrées et les
fronts actifs d'horloge.

Il est donc commode et recommandable de "caler" toutes les transitions des entrées sur les temps
multiples de la période d'horloge, et de décaler le générateur d'horloge d'une fraction de période.
Pour rendre plus lisible les signaux internes et éventuellement les sorties (machine de Mealy), on
placera le front actif d'horlogen peuavantles multiples de la période.

Dans cet exemple, I'horloge de périqulest calée de maniére a ce que son premier front actif arrive
alinstantp/4 + p/2 soit 3/4 de période. Les fronts actifs seront donc en avance de 1/4 de période
par rapport aux transitions des entrées.

On a utilisé l'instructiorparameter  pour créer ursymbolep représentant la période d'horloge.

© J.L.N.2005

Exemple : simulation du registre & décalage a chargement paralléle :

module test;

reg [3:0] data_in; // reg vectoriel de 4 bits
reg ck, load; wire ser_data;

parameter p=5000; // periode horloge

piso4 piso ( ck, load, data_in, ser_data ); // instance

initial //============= sequence principale
begin
load = 1; data_in = 1,
#p load =0;
#(p*6) load = 1; data_in = 10;
#p load = 0;
#(p*6) load = 1; data_in ='b1101,
#p load =0;
#(p*6) $finish;
end

initial  // ============= generation horloge
begin
ck = 0;
#(p/4) forever #(p/2) ck = ~ck;
end

always @ (posedge ck) // ===== affichage echantillonne
$display("ld=", load, " data_in=%b", data_in,

" ser_data=", ser_data );

endmodule

el ey e ey

load ] ] [ ]
data_in [ 1010 [1101
ser_dat F_‘ ’_| |_| | | | E—

© J.L.N.2005

Sortie texte : on a mis dans cet exemple un affichage de résultats échantillonné, produisant une ligne
de texte par période d’horloge : un blatways qui appelle la fonctioisdisplay  a chaque front

actif d'horloge. Dans le cas d'un systeme synchrone, ceci facilite l'interprétation des résultats
(contrairement a la fonctioBmonitor  qui produit un nombre variable de lignes par période).

De cette maniére, on "voit" les sorties du circuit exactement comme un autre module synchrone les
"verrait".

3. Déclarations

ATTENTION : le langagelistingueles majuscules et minuscules.

Les regs et nets doivent étre déclareés :

» alintérieur des modules, mais

» al'extérieur des blocs procéduraux

La déclaration des regs est toujours obligatoire.

La déclaration des nets est obligatoire sauf dans un cas : dans une représentation structurelle, les
nets internes du module qui apparaissent dans les instances sans avoir été déclarés explicitement
sont considérés comme déclarés implicitement (avec le type de net "par défaut”).

Attention Danger: il faut déclarer explicitement les nets vectoriels, car la déclaration implicite leur
donne une capacité de 1 bit seulement, sans avertissement.

Les ports du module sont des nets, et les déclarations de direction des s, (Qutput, inout

valent comme déclarations de nets.

Les types de regs sont les suivants :

* reg "simple” ou "scalaire" de 1 bit

* reg vecteur (nombre de bits arbitraire) (non-signé)
» table de regs vectoriels = "RAM"

* integer (reg de 32 bits, signé)

Il y a de nombreux types de nets, qui different principalement par la maniere dont sont résolus les
conflits et les indéterminations. Le type de net le plus approprié pour les circuits CMOS statiques
est lewire, qui résout les conflits en fonction des forces des sources en présence (cf 86) et prend

I'étatZ (haute impédance) lorsqu'aucune force ne détermine son état.

N.B. Le motnetn'est pas un nom de type, mais de "classe de types".

© J.L.N.2005

exemple de déclaration :
reg [15:0] a, b; reg c; reg [1:200] bizar;
reg [7:0] babar[0:127];
wire [23:0] leBus;
a etb sont des regs vectoriels de 16 bits, donskeporte I'indice 0,
C est un reg scalairdjzar est un reg de 200 bits dont &b porte l'indice 200,
babar est un tableau de 128 mots de 8 bits, pouvant étre utilisé comme modeéle d'une mémoire
RAM ou ROM, leBus est un net de 24 bits.

4. Expressions et Constantes

Les régles de constructions des expressions sont similaires dans les mondes concurrent et
procédural. Elles sont fortement compatibles avec le langage C.
N.B. En cas de doute sur la priorité des opérateurs, utiliser des parenthéses.

4.1 opérations logiques “verticales” ou “bitwise”

commeenC :
opérateurs ~, &, |, " pour respectivement not, and, or, xor
exemple :

c=aé&hb;

d=~c|~(a"b);
chaque bit de ¢ sera le "et" du bit de méme poids pris sur a et sur b.
L'opérateur ~ est prioritaire sur les autres.

4.2 opérations logiques “horizontales” ou “unary”

opérateurs &, |, »  pour respectivement and, or, xor
exemple :

reg [7:0] b; reg azero;

azero=~(|b);
azero est a 1 si et seulement si tous les 8 bits de b sont nuls.

4.3 opérations arithmeétiques et décalage

comme en C :
opérateurs +, -, *, /[, %, <<, >>
Attention : les opérateurs d'incrémentation et de décrémentation (++, --) n'existent pas en Verilog.

© J.L.N.2005

4.4 opération booléennes

commeen C:

opérateurs !, &&, || pour respectivement not, and, or

étant entendu que la valeur zéro représéfaise” et que toute valeur différente de zéro représente
"true".

4.5 opérations relationnelles

commeen C:

opérateurs ==, =, <, > <=, >= pour respectivement égal, différent, inférieur, supérieur,
inf. ou égal, sup. ou égal.

avec en plus :

opérateurs ===, !== pour respectivement littéralement égal, littéralement différent.

(au sens de= et !=, I'état X implique un résultat de comparaison toujours égal a X, alors que au
sens dee== et!==, X est égal a lui-méme)

N.B. les comparaisons sont non signées, sauf pour le type integer.

4.6 vecteurs

Les opérateurs{ et } servent a former un vecteur par concaténation de bits ou de vecteurs
(msb=most significant bien premier).
exemple :
reg [8:0] resu; reg [2:0] rev_a; reg g0, q1, g2;
wire [7:0] sum; wire [2:0] a;
resu = { carry, sum[7:0] };
{d2,q1,q0}=a+b;
rev_a = {a[0], a[1], a[2] };
c=sum[n-1];
resu est un vecteur de 9 bits dont le msb eatry
g2, ql, g0 recoivent séparément les 3 Isb de la sonamieb,
rev_a contient les trois Isb da dans l'ordre inverse,
c est len-ieme bit desum (n étant un entier >=1).

4.7 constantes numériques

Les constantes numeériques sont décimales par défaut.
Les opérateurh, 'h ,'o préfixent les constantes binaires, hexadécimales et octales.
(ces opérateurs peuvent étre précédés d'un entier indiquant le nombre de digits.)

Le caractere ("souligné™) peut étre inséré librement pour rendre la constante plus lisible.

La déclaratiorparameterpermet d'associer un symbole a une constante.
© J.L.N.2005

exemples :

parameter msb8 = 'b1000_0000;
a=a+ 'hFCO0;
b ='bz; /I etat haute impédance

4.8 extension - troncature

- si un élément d'une expression comporte plus de bits que la destination de cette expression, cet
élément va étre tronqué "par la gauche", c'est a dire que ce sont les bits de faible poids qui vont

subsister.

- si un élément d'une expression comporte moins de bits que la destination de cette expression, cet
€lément va étre complété par des zéros "a gauche" (du coté des forts poids).

Attention : ces opérations ne s'accompagnent d'aucun avertissement et peuvent dissimuler des
erreurs fatales.

5. Communication reg <--> net

Régles fondamentales :

- I'état d'un reg n'est modifié que par assignement procédural (une instruction exécutée de maniere
séquentielle). L'assignement procéduraliestantané le reg conserve son état jusqu'au prochain
assignement.

- I'état d'un net n'est modifié que par connexion structurelle ("branchement” d'un instance) ou par
assignement continu ("branchement” d'un net a une expression). Dans ces deux cas, l'action sur le
net estpermanente

Conséquences :

- un reg ne peut pas étre branché a un port de sortie d'une instance, ni a un port bidirectionnel
(inout),

- un net ne peut pas modifié directement dans une séquence d'instructions constituant un modele
procédural.

Transmission net -->reg :
ne présente aucune difficulté dans un bloc procédural, car le membre de droite d'un assignement
procédural peut étre une expression contenant librement regs et nets.

Transmission reg --> net :

- si le net est en dessous dans la hiérarchie structurelle, la transmission peut se feirengaon

du reg a urportd'entrée d'une instance.

C'est ainsi qu'on procéde en général pour injecter les stimuli dans le module a tester : voir pour
exemples 82.1.3 et §2.3.

- si le net est au dessus ou au méme niveau dans la hiérarchie structurelle, la transmission peut se
faire par assignementontinu (le membre de droite d'un assignement continu peut étre une
expression contenant librement regs et nets).

C'est ainsi qu'on procede en général pour acheminer le résultat du modéle procédural d'un module
vers les ports de sortie de ce module : voir pour exemples §2.2.1, §2.2.2 et 82.2.3.

(On en profite souvent pour introduire dans cet assignement le temps de propagation des sorties du
module).

N.B. Ne pas confondre assignement procédural et assignement continu.

L'assignement procédural est dans un bloc procédural, il ne contient pas le mot-clef "assign" mais
seulement le signe "=", son membre de gauche est un reg.

L'assignement continu egh dehors de tout bloc procédural, il commence par le mot-es$ign,

son membre de gauche est un net.

6. Etats et Forces

Le simulateur Verilog connait quatre états logiques : 0, 1, X et Z.

L'étatX est I'état "indéterminé" ou "inconnu"”.
C'est une pure abstraction : il n'existe pas dans le circuit réel, mais sert a traduire une absence de
déterminisme a un instant particulier de la séquence de simulation.
C'est I'état initial de tous les regs et nets. C'est aussi I'état qui apparait lorsqu'un conflit entre 1 et O
ne peut étre résolu.
Cet état n'est pas considéré comme une erreur, mais comme un état a part entiere qui est propagé
par application d'une extension de l'algebre de Boole :
1&X-->X 1| X-->1 ~X --> X
0&X-->0 0| X-->X
En sortie, il sert a détecter un manque d'initialisation ou un conflit.
En entrée, il peut étre utilisé pour faire une analyse de déterminisme : exemple :
Supposons qu'on veuille prouver que lorsque lapii de I'entréea[7:0]  d'un circuit est nul le
résultat est indépendant des 7 autres bita.d@our cela il suffit de faire :
a = 'bOXXX_XXXX;
et d'observer que aucune des sorties n'est a X (on évite ainsi de devoir essayer 128 valpurs de

L'état Z est I'état haute-impédance. Il traduit un phénomeéne physique réel : le noeud "flottant". I
apparait lorsqu'une entrée est "en l'air" ou lorsqu'un bus bidirectionnel n'est contrélé par aucun des
drivers tri-state qui y sont connectés.

Lors de I'évaluation des expressions logiques, il est d'abord converti en état X.

Les conflits apparaissent lorsque plusieurs sources de signal (sorties ou ports bidirectionnels ou
assignements continus) sont connectées sur le méme net, et tendent a le forcer a des états
contradictoires.

Bien entendu, I'état Z ne contredit aucun autre vu son caractere passif.

(Mais I'état X contredit tous les autres).

Une échelle déorcespermet de contréler la résolution des conflits.

Si les sources ont des forces différentes, la plus forte impose sa valeur, si elles ont des forces égales,
la contradiction se traduit par I'état X du noeud.

Echelle des 5 forcessupply, strong, pull, weak, highz.

La force par défaut eststrong

Exemple :
Modélisation de portes NMOS ou DCFL-MESFET ou "open-collector" avec pull-up passif :

module DCFL_nor2(s, a, b);
output s; input a, b;
assign (strongO, pulll) #(2.5,2.0)s=~(a|b);
endmodule
En "court-circuitant” les sorties de plusieurs de ces portes, on effectue un "et cablé" puisque les

conflits se résolvent en faveur du 0.

Autre exemple : forcage des noeuds d'un circuit pour initialisation :

On ajoute un assignement continu (conflictuel) sur un noeud déja connecté a une sortie de cellule,
avec une forcasupply  supérieure a la force par défaut.

Pour relacher le forcage, on donne a cette source la valeur Z.

7. Blocs Procéduraux

7.1 Regles

La modélisation procédurale est contenue dans des blocs procéduraux (ou dans les taches et
fonctions appelées par ces blocs). Les blocs procéduraux sont des portions de programme qui sont
exécutées en paralléle.

Il n'y a pas de hiérarchie des blocs procéduraux, un bloc ne peut pas en appeler un autre.

(Par contre un bloc procédural peut appeler des taches et fonctions).

Les blocs procéduraux ne sont pas appelés, mais démarrent tous simultanément a l'instant zéro.

Les blocs procéduraux sont obligatoirement contenus dans des modules.

L'ordre de leurs déclarations n'a aucune incidence sur leur exécution.

lls ne contiennent ni instances, ni assignements continus.

Exemples :
bloc initial (exécuté une seule fois) : voir §2.1.3
bloc always(réitéré indéfiniment) : voir §2.2

7.2 Temporisation procédurale

Les conditions de temporisation procédurdle@ et wait peuvent apparaitre soit dans l'en-téte
d'un blocinitial ou always , soit en préfixe d'une instruction dans un bloc

7.2.1 délai relatif

Le caractére # suivi d'un nombre ou d'un expression numérique introduit une attente avant
I'exécution d'une instruction ou d'un bloc (et retarde d'autant toute la suite du bloc).

On ne peut spécifier le temps de maniere absolue (mais en cas de besoin, on peut créer un bloc
initial contenant seulement linstruction dont on veut spécifier l'instant d'exécution de fagon
absolue.)

7.2.2 attente de transitions ou de front

Le caractere @ suivi d'une liste de signaux entraine l'attente de la prochaine transition d’'un des
signaux de la liste. Sous cette forme, cette condition permet la modélisation de circuits
combinatoire de maniére procédurale, exemple :

always @ (AorBorC)
begin

Chaque fois que l'un des signaux d'entrée A, B, ou C change d'état, la sortie RESU est re-calculée
(comparable au "process” de VHDL).

Avec les mots-clefposedge (front montant) enegedge (front descendant) on peut choisir quel
front va étre attendu, ce qui nous fournit I'outil de base de modélisation des systgmbgeones:

@ (posedge clk)
attendra le prochain front actif d'horloge.

7.2.3 attente d’un niveau

Le mot-clef wait suivi d'un expression logique va faire attendre que cette expression prenne une
valeur "vraie" (non-nulle). Si I'expression est vraie au moment de I'exécution, il n'y a pas d'attente.

N.B. ne pas confondrevait etif : wait bloque I'exécution tant que la condition n'est pas
satisfaite, alors qué saute immédiatement a l'instruction suivante.

8. Constructions Procédurales

A l'intérieur d'un bloc procédural, on trouve des instructions simples ou composées, éventuellement
préfixées de conditions de temporisation.

Une instruction simple peut étre :

* un assignement procédural

* un appel de tache

* une construction itérative (boucle) ou décisionneali@|( cass.

Dans les deux premiers cas, elle doit étre terminée par le caréctere

Une instruction composée est une séquence d'instructions encadrée par les molkeglafs et
end.

Les constructions sont trés similaires au langages C, a deux difféerences pres :

» présence des conditions de temporisations,

* begin etend a la place d§ et}.

N.B. Il n'y a pas de danger a mettre plus de coupbegin - end qu'il n'est strictement
nécessaire, et cela peut améliorer la lisibilité du texte (exemple § 2.2.1)

8.1 Itérations ou boucles

8.1.1 boucle “for”

similaire au C :
for ( <assignement>; <condition>; <assignement> )
<instruction>

exemple :
for (i=0;i<IMAX;i=i+l)
begin
@ (posedge clk) a=a>> 1,
#llx=a &1,
end

N.B. la variable de contréle doit avoir été déclarée (integer en général)

8.1.2 boucle “while”

similaire au C :
while ( <condition>)
<instruction>

exemple :
while (a[7] == 0) // justification a gauche de a[7:0]
a=-a<<l],;

”

8.1.3 boucle “repeat

repeat ( <expr>)
< instruction >

exemple :
repeat (3) begin
a=1;
#ta=0;
#(t*3) ; //le";" représente une inst. vide
end

La variable qui sert a compter les "tours de boucle" est cachée, on ne peut ni la déclarer ni y accéder
explicitement.

8.1.4 boucle “forever”

forever <instruction>
exemple

forever #(p/2) clk = ~clk;

Attention : Il faut impérativement éviter une boucle capable de faire un nombre illimité de "tours"
en un temps de simulation nul. Cditoque irrémédiablement la simulation sans que l'utilisateur soit
averti.

Exemple :

while (a == 0) begin end // attendre que a passe a 1
blogue la simulation sans attendre qupasse a 1, doit étre remplacé par

wait (a ) ;
Autre exemple : I'exemple du 8§ 8.1.2 va bloquer la simulation si a[7:0] est nul. Si cela risque de se
produire, il faut modifier le programme comme suit :

1=7;
while ( (a[7]==0) && (i) ) //justification & gauche
begin a=a<<1;i=i-1; end

8.2 Décisions
8.2.1 décision “if”

similaire au C :
if ( <condition>) <instruction>
else <instruction>

ou simplement :

if ( <condition>) <instruction>
exemple : cf§2.2.1

8.2.2 expression conditionnelle

similaire au C :
( <condition>) ? <instruction> : <instruction>
N.B. peut aussi s'utiliser dans un assignement continu !

exemple:cf§9.2et§10.1.1

8.2.3 décision “case”

trés différent du C !

case ( <expression>)
<constante> : <instruction>

default : <instruction>

endcase
exemple :
case (opcode) /l ALU combinatoire
l:s=aé&hb;
2:s=a]lb;
5:s=a+b;
7:.s=-~a,
default : s = HXXXXXXXX  XXXXXXXX;
endcase

9. Encore quelques exemples

9.1 bascule D avec clear asynchrone

Soit a faire le modeéle d'une bascule D sensible au front montant, avec une entrée de remise a zéro
asynchrone (prioritaire) sensible au niveau :

module LibDFFc ( q, ck, d, clear);
input ck, d, clear; output q;
reg iq;
always @ ( posedge ck) // <== sensible au front
if (clear==0)iq=d,;
always wait ( clear == 1) // <== sensible au niveau
begin
iq = 0;
wait (clear==0) ;
end
assign #8 q = iq;
specify // timing a verifier : setup 5.2 ns, hold 1.5 ns
$setuphold( posedge ck, d, 5.2, 1.5);
endspecify

endmodule

N.B. Si on oublie la conditiorwait ( clear == 0 ) a la fin du second bloc always, celui-ci

va boucler un nombre indéfini de fois dés que clear sera a 1 ==> simulation bloquée (cf § 8.1).
Remarquer aussi que cette condition préfixe une instruction vide, représentée pdr un
(obligatoire).

Le bloc "specify" sert ici a demander au simulateur des vérifications de "timing", dans le cas présent
il devra signaler par un "warning" toute transition dl€ui surviendrait entre 5.2 ns avant et 1.5 ns
apres le front montant d'horloge.

9.2 Buffer tri-state

module LibTRI (out, in, out_ena);
inout out; input in, out_ena;
assign (strong0, highz1l) #1 out =in | ~out_ena;
assign (highz0, strongl) #1 out =in & out_ena;
endmodule
variante :
module LibTRI (out, in, out_ena);
inout out; input in, out_ena;
assign #1 out = (out_ena) ?in : 'bz;
endmodule

9.3 Cellule paraméetrée

L'utilisation de paramétres permet de faire des descriptions “génériques”.
exemple : le registre a décalage du § 2.2.1 généralid bits :
/I registre piso de N bits
module pisoN( clk, load, par_in, ser_out );
parameter N=4;
input clk, load; input [(N-1):0] par_in;
output ser_out;
reg [(N-1):0] contenu;
always @ (posedge clk)
contenu = (load)?(par_in):(contenu >> 1);
assign #3 ser_out = contenul0];
endmodule

exemple d’instance d’un piso de 16 bits :
defparam piso_inst.N = 16;
pisoN piso_inst ( ck, load, data_in, ser_data );

La directive defparam a permis de fixer une valeur de N différente de la valeur par défaut.

10. Introduction a la Synthese Logique

C'est la technique qui consiste a utiliser un logiciel de synthése pour obtenir automatiquement une
description structurelle (utilisant les cellules d'une bibliotheque préalablement définie) a partir d'une
description comportementale.

(Accessoirement cette technique peut aussi étre utilisée pour optimiser une description structurelle
existante, ou I'adapter a une bibliotheque différente ("reciblage").)

Il faut savoir qu'il ne suffit pas qu'une description comportementale donne un résultat correct a la
simulation pour qu'elle soit "synthétisable”, il y a de nombreuses restrictions dans I'emploi des
primitives du langage. De plus un logiciel de synthése utilisé sans précautions est capable de
produire un circuit N fois plus compliqué gu'il n'est nécessaire.

On donnera ici quelques directives générales, en précisant bien que chaque logiciel de synthese
impose des contraintes particulieres.

10.1 Modules combinatoires

10.1.1 description concurrente

La solution "sans problémes" est la description concurrente a e,
Elle s’applique spécialement bien aux cas ou on dispose déja d’expressions booléennes (non
nécessairement minimales) des fonctions a réaliser : exemple : le multiplexeur du 8§ 2.2.3.
module LibMux( 10, i1, s,y );
input i0, i1, s; output y;
assign#1ly=(i0&~s) || (i1l &s);
endmodule

On peut aussi exploiter d’autres possibilités des expression Verilog :
module LibMux(i0, i1, s,y );
input i0, i1, s; output y;
assign #1 y = (s)?(i1):(i0);
endmodule

variante (généralisabla N entrées) :

module LibMux( ivec, s,y );
input s; input [1:0] ivec; output y;

assign #1 y = ivec|[s];
endmodule

autre exemple : additionneur-soustracteur 8 bits :
module addsous( A, B, sign, X);
input [7:0] A, B; input sign; output [8:0] X;
assign X = (sign)?(A+B):(A-B);
endmodule © J.L.N.2005

10.1.2 description procédurale

Si la spécification du bloc combinatoire se présente sous forme d’'une table de vérité, on peut
vouloir éviter I'extraction manuelle des expressions booléennes des sorties.
Pour décrire le module directement sous forme de table, on peut utiliser la décision “case” (voir
§ 8.2.3) dans un bloc procédural déclenché par les transitions des entrées :

module tableau( A, B, Z);
input [2:0] A; input B; output [4:0] Z;
reg [4:0] rZ;
always @ (AorB)
case ({A,B})
'b0001 : rZ ='b10001;
'b0100 : rZ ='b01001;
'b0110 : rZ ='b01111;
default : rZ ='b10000;
endcase
assign Z =rZ,
endmodule

N.B : on a di0 déclarer un reg pour la sortie de la table. Le reg ayant en général une capacité de
mémorisation, le programme de synthése risque de mettre des bascules, ce qui est contraire a
I'objectif de faire un circuit combinatoire.

Heureusement I'effet de mémorisation est évité si on respecte :

Les deux conditions de modélisation procédurale d’'un bloc combinatoire :

 toutes les variablesntervenant dans I'évaluation du reg apparaissent dans la condition de
déclenchement du bloc always,

* le reg est re-évalué damsus les cagd’éxécution du bloc

Dans I'exemple ci-dessus, la clause default du case permet de garantir la seconde condition.

D’une facon plus générale, un circuit combinatoire peut étre décrit par une construction procédurale
si les deux conditions ci-dessus sont satisfaites.
Un autre exemple : supposons qu'on veuille faire un module comptant les "uns" du vecteur d'entrée :
module compte_uns(in, res);
input [2:0] in; output [1:0] res;
reg [1:0] r; integer cnt;
always @ (in)
begin
r=0;
for(cnt=0;cnt<3;cnt=cnt+1)
r=r+infcnt];
end
assignres =r;

endmodule

Bien que l'algorithme soit séquentiel, le circuit synthétisé par logi€ghopsyssera purement
conbinatoire.

"""""""" e
- o]

""" =
ok p ST

10.2 Modules Séquentiels Synchrones

La description fait appel a un bloc procédural, de type

always @ (posedge clk)
décrivant les actions a effectuer entre deux fronts actifs d'horloge.

Ce bloc ne doit contengiucune temporisationautre que celle de I'en-téte.

Ce bloc constitue la boucle unique de la description, il ne doit pas en général contenir de boucle
interne genre “while” ou “for”.

Les signaux de typeeg étant doués de mémoire, le logiciel va avoir tendance a placer une bascule
pour chaque bit deeg recevant des valeurs a l'intérieur du bloc always.

Il faut donc soigneusement évaluer les besoins en mémoire du systéme projeté et ne déclarer que le
minimum dereg s nécessaires.

exemple : le registre du § 2.2.1 est décrit dans un style parfaitement synthétisable.

Il est également possible gu'un méme module contienne plusieurs blocs procéduraux, dont un seul
déclenché par le font d’horloge, et un ou plusieurs autres représentant des parties combinatoires
respectant les deux conditions expliquées au § 10.1.2.

Attention : un méme reg ne doit pas subir des affectations dans deux blocs procéduraux ayant des

conditions de déclenchement différentes. Ceci aboutirait ampessibilité de synthése.
(Ainsi I'exemple du 8 9.1 n’est pas synthétisable)

10.3 Machine a états finis

Considérons une machine a états finis ou automate synchrone selon le modéle “de Moore” et
décrivons la dans le style “RTL” :

MACHINE A ETATS FINIS (FSM) (''de Moore")

L DFF

Logique / \ / \ Logique
Combi- Combi—
natoire

natoire \

p—
[
 pp—

clk

Entrees
Ftat
FUTUR

PRESENT
Sorties

Ftat

module moore_fsm( entrees, sorties, clk );
input [1:0] entrees; input clk; output [3:0] sorties;
reg [2:0] etat_futur, etat_present;
reg [3:0] r_sorties;
always @ (posedge clk)

etat_present = etat_futur;
always @ (etat_present or entrees)

begin

< calcul etat_futur >

end
always @ (etat_present)
begin
< calcul r_sorties >
end
assign sorties = r_sorties;
endmodule

Naturellement il est possible de remplacer le second ou le troisieme bloc procedural par une
description concurrente (& basessign3.

Variante ou I'état futur n’apparait pas explicitement :

module moore_fsm( entrees, sorties, clk );

input [1:0] entrees; input clk; output [3:0] sorties;
reg [2:0] etat_present;

reg [3:0] r_sorties;

always @ (posedge clk)

begin
< calcul etat_present >
end
always @ (etat_present)
begin
< calcul r_sorties >
end
assign sorties = r_sorties;
endmodule

10.4 Registre avec clear asynchrone

La description de bascule du 9.1 est correcte mais n’est pas acceptée par certains programmes de
synthese. On utilise alors la variante suivante (qui n’est pas strictement correcte) :

always @ ( posedge ck or posedge clear)
if (clear)
iq=0;
else ig=d;

De cette maniére on peut facilement introduire le clear asynchrone dans une description de type
RTL comme celle du § 10.3.

11. Ce qui resterait a voir

Ce document n'est pas le manuel de référence du langage Verilog mais seulement une introduction.
Il faut savoir qu'il existe de nombreuses possibilités complémentaires...

» primitives du simulateur, soit modules pré-définis représentant les portes logiques de base
» primitives définies par l'utilisateur (UDP) par représentation tabulaire (comb. et seq.)

* modélisation MOS au niveau transistor, modélisation MOS dynamique (effets capacitifs)

» échelle des 3 cas de temps de propagation : min, typ, max

* modeles de propagation (inertiel et transport)

» "path delay specify" : spécifications des temps de propagation par chemins

* assignement continu incorporé a la déclaration de net

» délais, forces incorporés a la déclaration de net

» types de net autre que wire

* instances avec ports non connectes

» conflits de types de net aux ports d'une instance

* output port déclaré comme reg --> assignement continu implicite

* nombres réels

» "command line options" : options d'invocation, organisation des bibliotheques ("libraries™)

* mode interactif et mise au point ("debugging”, "patching™)

» "tasks", "functions" : taches et fonctions définies par I'utilisateur (= sous-programmes)
« fonctions et taches pré-définies autres que celles utilisées dans les exemples ($xxxx)
» acces a des fichiers de données

» directives de compilation "default_nettype, "define, ‘'unconnected_drive, etc..

* événements nommeés ("named events")

» parallélisme a l'intérieur d'un bloc procédural (“fork", "join™)

* "intra-assignement timing" : délai a l'intérieur d'un assignement procédural

